# $Id: PKGBUILD 55453 2011-09-12 23:51:53Z ebelanger $
# Contributor: Jaroslaw Swierczynski <swiergot@aur.archlinux.org>
# Contributor: Sven Kauber <celeon@gmail.com>
# Maintainer: Daniel J Griffiths <ghost1227@archlinux.us>

pkgname=rlog
pkgver=1.4
pkgrel=4
pkgdesc="A flexible message logging facility for C++ programs and libraries"
arch=('i686' 'x86_64')
url="http://www.arg0.net/rlog"
license=('LGPL')
depends=('gcc-libs')
options=('!libtool')
source=(http://rlog.googlecode.com/files/${pkgname}-${pkgver}.tar.gz \
	${pkgname}-section.patch)
md5sums=('c29f74e0f50d66b20312d049b683ff82'
         '69644989e07ceb5c265dfc0c6f770810')

build() {
	cd "${srcdir}/${pkgname}-${pkgver}"
	patch -Np1 -i "${srcdir}/${pkgname}-section.patch"
	./configure --prefix=/usr
	make
}

package() {
	cd "${srcdir}/${pkgname}-${pkgver}"
	make DESTDIR="${pkgdir}/" install
}
